# Recovering sech solution with sech ansatz

```
> restart;
> interface(showassumed=0):;
```

Ansatz of the form u(x,t) = A(x,t) * exp(I*phi(x,t))

```
> A := a(t)*sech(a(t)*(x-xi(t)));
> phi := c(t)*(x-xi(t))+b(t);
```

$$A := a(t)\,\mathrm{sech}(a(t)\,(x-\xi(t)))$$
$$\phi := c(t)\,(x-\xi(t)) + b(t) \tag{1}$$

Lagrangian: (I/2)(u u*_t - u_t u*) + (1/2) u_x u*_x - (1/2) |u|^4 = A^2 phi_t + (1/2)*(A_x^2+A^2 phi_x^2) -(1/2)*A^4

```
> LA := A^2*diff(phi,t) + (1/2)*(diff(A,x)^2+A^2*diff(phi,x)^2) -
  (1/2)*A^4;
```

$$LA := a(t)^2\,\mathrm{sech}(a(t)\,(x-\xi(t)))^2\left(\left(\frac{d}{dt}c(t)\right)(x-\xi(t)) - c(t)\left(\frac{d}{dt}\xi(t)\right) + \frac{d}{dt}b(t)\right) + \frac{a(t)^4\,\mathrm{sech}(a(t)\,(x-\xi(t)))^2\,\tanh(a(t)\,(x-\xi(t)))^2}{2} + \frac{a(t)^2\,\mathrm{sech}(a(t)\,(x-\xi(t)))^2\,c(t)^2}{2} - \frac{a(t)^4\,\mathrm{sech}(a(t)\,(x-\xi(t)))^4}{2} \tag{2}$$

Take out t-dependence so we can compute Leff

```
> sub1 := (diff(c(t),t))=cp,(diff(a(t),t))=ap,(diff(b(t),t))=bp,
  (diff(xi(t),t))=xip:
> sub2 := xi(t)=xi,b(t)=b,c(t)=c,a(t)=a:
> LA2 := subs({sub1,sub2},LA);
```

$$LA2 := a^2\,\mathrm{sech}(a\,(x-\xi))^2\,(cp\,(x-\xi) - c\,xip + bp) + \frac{a^4\,\mathrm{sech}(a\,(x-\xi))^2\,\tanh(a\,(x-\xi))^2}{2} + \frac{a^2\,\mathrm{sech}(a\,(x-\xi))^2\,c^2}{2} - \frac{a^4\,\mathrm{sech}(a\,(x-\xi))^4}{2} \tag{3}$$

Leff = integral of Lag, we need to assume that a>0 and xi real to be able to evaluate integrals

```
> assume(a>0);
> assume(xi,real);
> Leff := int(LA2,x=-infinity..infinity);
```

$$Leff := -\frac{a\,(a^2 - 3\,c^2 + 6\,c\,xip - 6\,bp)}{3} \tag{4}$$

Put back t-dependences so that we can do Euler-Lag

```
> subi1 := cp=diff(c(t),t),ap=(diff(a(t),t)),bp=(diff(b(t),t)),xip=
  (diff(xi(t),t)):
> subi2 := xi=xi(t),b=b(t),c=c(t),a=a(t):
```

Euler-Lagrange eqs

```
> dLap := subs({subi1,subi2},diff(Leff,ap)):;
> dLbp := subs({subi1,subi2},diff(Leff,bp)):;
```

```
> dLcp := subs({subi1,subi2},diff(Leff,cp)):;
> dLxip:= subs({subi1,subi2},diff(Leff,xip)):;
> eq1 := diff(dLap ,t)=subs({subi1,subi2},diff(Leff,a));
> eq2 := diff(dLbp ,t)=subs({subi1,subi2},diff(Leff,b));
> eq3 := diff(dLcp ,t)=subs({subi1,subi2},diff(Leff,c));
> eq4 := diff(dLxip,t)=subs({subi1,subi2},diff(Leff,xi));
```

$$eq1 := 0 = -a(t)^2 + c(t)^2 - 2\,c(t)\left(\frac{d}{dt}\,\xi(t)\right) + 2\,\frac{d}{dt}\,b(t)$$

$$eq2 := 2\,\frac{d}{dt}\,a(t) = 0$$

$$eq3 := 0 = -\frac{a(t)\left(-6\,c(t) + 6\,\frac{d}{dt}\,\xi(t)\right)}{3}$$

$$eq4 := -2\left(\frac{d}{dt}\,a(t)\right)c(t) - 2\,a(t)\left(\frac{d}{dt}\,c(t)\right) = 0 \qquad \textbf{(5)}$$

Sove Euler-Lag ODEs simultaneously

```
> sol:=solve({eq1,eq2,eq3,eq4},{diff(c(t),t),diff(a(t),t),diff(b
  (t),t),diff(xi(t),t)}):
> sol[1];sol[2];sol[3];sol[4];
```

$$\frac{d}{dt}\,a(t) = 0$$

$$\frac{d}{dt}\,b(t) = \frac{a(t)^2}{2} + \frac{c(t)^2}{2}$$

$$\frac{d}{dt}\,c(t) = 0$$

$$\frac{d}{dt}\,\xi(t) = c(t) \qquad \textbf{(6)}$$